Parameter Estimation Example

Kurt Bryan and SIMIODE

A very simple example of fitting a function or model to data by using least squares.

```
> restart;
  with(plots) :
```

The Data: Here are some hypothetical data in the form of (t,y) pairs:

```
> data := [[1.1, 1.24], [1.9, 0.83], [2.3, 0.71], [4.1, 0.29], [5.5, 0.15]]
```

A plot of the data:

```
> plt1 := pointplot(data, color = red, symbol = solidcircle, symbolsize = 25, labels = ["t", "y"])
```

The Model and Sum of Squares: Let's fit a model f(t) = a*exp(b*t) to this data by adjusting a and b. First form a sum of squares

```
> f(t) := a·exp(b·t);
  SS := add( (f(data[j][1]) − data[j][2])^2, j = 1 ..5)
```

With only two parameters "a" and "b", a visual estimate of the best choice (the choices of a and b that minimize SS) can be found by plotting. It's clear that b<0 since the data decays, and also that a>1

```
> plot3d(SS, a = 1 ..3, b = -1 ..0)
```

Or perhaps plotting the log reveals more information

```
> plot3d(ln(SS), a = 1 ..3, b = -1 ..0)
```

Rotate the graph around. Something around a = 2, b = -0.5 looks promising.

Minimizing the Sum of Squares: The multivariable calculus approach is to find a critical point. Form the appropriate derivatives

```
> dSSda := diff(SS, a);
  dSSdb := diff(SS, b);
```

and solve for a and b, numerically, with an appropriate initial guess

```
> absols := fsolve({dSSda = 0, dSSdb = 0}, {a = 2, b = -0.5})
```

The residual sum of squares is

```
> evalf(subs(absols, SS))
```

A plot the best-fit f(t) to compare to the data:

```
> plt2 := plot(subs(absols, f(t)), t = 1.1 ..5.5, color = blue) :
  display(plt1, plt2)
```

Alternatively, we can minimize SS with respect to a and b by using Maple's built-in optimization routines. First load the ***Optimization*** package:

```
> with(Optimization) :
> minsol := Minimize(SS, initialpoint = {a = 2, b = -0.5})
```

The residual is the first output.

```
>
```